Dr Marcin Kolasa
Advanced Macroeconomics
Warsaw School of Economics

Introduction to Dynare for Octave

Installing and setting path

Download Dynare from http://www.dynare.org/download/ and instal it. In order to let Octave have access to Dynare codes automatically after it starts, add to file `.octaverc` a line setting the path to folder `matlab` in the Dynare installation folder.
For example, the appropriate line in `.octaverc` may look like this:

```
addpath C:\dynare\4.1.2\matlab;
```

MOD file

The model that we want to solve and simulate in Dynare should be coded in a text file with an extension `MOD`. Such a file must have an appropriate syntax. The main components of a MOD file are as follows (in some cases, the sequence matters):[1]

- Command `var`: it should be followed by the names of all endogenous variables, separated by spaces or commas (this also applies to the following two commands)
- Command `varexo`: serves to declare exogenous variables, i.e. shocks; IMPORTANT: if shocks are driven by AR (ARMA, VAR, VARMA etc.) processes, they are treated as endogenous variables, while innovations to these processes (which are IID) are treated exogenous variables
- Command `parameters`: declares names of parameters
- Numerical values of parameters
- Model equations: we enter them within a block starting with command `model;` and ending with command `end;` Lags and leads should be mentioned in round brackets following a given variable, e.g. `x(-3)` is third lag of x, while `x(+2)` is the expected value of x two periods ahead, given the currently available information. The model equations can be written in their original form or log-linearized. In the former case, however, writing a separate code finding the steady state may turn out to be necessary (finding the steady state solution of a log-linearized model is trivial). In what follows we will assume that the model is written in log-linearized form. IMPORTANT: The timing of each variable reflects when that variable is decided (a typical example is capital stock in the standard RBC model: since it is decided in the previous period, its current value should be coded as `k(-1)`, not `k`)
- Command `steady;` means that the steady state equilibrium will be the starting point for any simulation (an alternative starting point can be given within the `initval;` … `endval;` block)
- Command `check;` Checks the Blanchard-Kahn and rank conditions
- Shocks: Shock characteristics should be entered within a block starting with command `shocks;` and ending with command `end;` For instance, the variance of shocks is declared using command `var`.
- Command `stoch_simul`: starts solving and simulating the model. Its options (e.g. setting the length of the impulse responses: `irf=100`) are entered in round brackets following the command and separated with commas.
- A `MOD` file, implementing the standard RBC model with endogenous labour supply can look as follows:

[1] Więcej szczegółów w dokumentacji Dynare, wgrywanej automatycznie do katalogu doc w folderze instalacji pakietu.

```
var y c l k A;

varexo eps;

parameters alpha beta delta theta rho sigma varphi;

alpha=0.33;
beta=0.99;
delta=0.025;
theta=1;
rho=0.95;
sigma=0.008;
varphi=1;

model;
y=A+alpha*k(-1)+(1-alpha)*l;
k=(1-delta)*k(-1)+(1/beta-1+delta)/alpha*(A+alpha*k(-1)+(1-alpha)*l-c)+delta*c;
varphi*l=-theta*c+A+alpha*k(-1)-alpha*l;
-theta*c=beta*(-theta/beta*c(+1)+(1/beta-1+delta)*(A(+1)-(1-alpha)*k+(1-alpha)*l(+1)));
A=rho*A(-1)+eps;
end;

steady;
check;

shocks;
var eps=sigma^2;
end;

stoch_simul(irf=100);
```

Dynare output

To run a given model in Dynare, one needs to type `dynare nazwa_modelu` in the command line of Octave, where `nazwa_modelu` is the name of the appropriate MOD file saved in the current folder of Octave.

The main components of the Dynare output are:

- STEADY-STATE RESULTS: steady state solution of the model (if the model is log-linearized, these are just zeros)
- EIGENVALUES: eigenvalues from the Jordan or Schur decomposition;[2] info whether the Blanchard-Kahn and rank conditions are satisfied
- POLICY AND TRANSITION FUNCTIONS: model solution in recursive form
- THEORETICAL MOMENTS: theoretical (ergodic) means, standard deviations and variances of model variables
- MATRIX OF CORRELATIONS: matrix of contempraneous correlations between model variables
- COEFFICIENTS OF AUTOCORRELATION: autocorrelation coefficients for model variables (several orders)
- Plots showing impulse responses to shocks

[2] See course material „Solving DSGE model using the Blanchard-Kahn algorithm".

The Dynare output for a model coded in the `MOD` file above is as follows:

```
STEADY-STATE RESULTS:
y            0
c            0
l            0
k            0
A            0

EIGENVALUES:
       Modulus           Real                  Imaginary

         0.95           0.95                        0
       0.9531         0.9531                        0
         1.06           1.06                        0
    4.085e+016     -4.085e+016                      0
    1.291e+017     -1.291e+017                      0

There are 3 eigenvalue(s) larger than 1 in modulus
for 3 forward-looking variable(s)

The rank condition is verified.

MODEL SUMMARY
  Number of variables:        5
  Number of stochastic shocks: 1
  Number of state variables:   2
  Number of jumpers:           3
  Number of static variables:  1

MATRIX OF COVARIANCE OF EXOGENOUS SHOCKS
Variables    eps
eps        0.000064

POLICY AND TRANSITION FUNCTIONS
                    y             c             k             l             A
k(-1)          0.217624      0.553075      0.953146     -0.167725             0
A(-1)          1.243572      0.367237      0.102394      0.438168      0.950000
eps            1.309024      0.386565      0.107783      0.461229      1.000000


THEORETICAL MOMENTS
VARIABLE       MEAN      STD. DEV.  VARIANCE
y             0.0000      0.0401     0.0016
c             0.0000      0.0303     0.0009
k             0.0000      0.0410     0.0017
l             0.0000      0.0088     0.0001
A             0.0000      0.0256     0.0007

MATRIX OF CORRELATIONS
Variables    y         c         k         l         A
y          1.0000  0.9134  0.8180  0.7114  0.9865
c          0.9134  1.0000  0.9813  0.3636  0.8344
k          0.8180  0.9813  1.0000  0.1776  0.7128
l          0.7114  0.3636  0.1776  1.0000  0.8168
A          0.9865  0.8344  0.7128  0.8168  1.0000

COEFFICIENTS OF AUTOCORRELATION
Order    1         2         3         4         5
y       0.9653  0.9317  0.8990  0.8673  0.8366
c       0.9944  0.9870  0.9780  0.9675  0.9558
k       0.9988  0.9953  0.9899  0.9826  0.9737
l       0.9065  0.8198  0.7393  0.6647  0.5956
A       0.9500  0.9025  0.8574  0.8145  0.7738
```

Accessing Dynare output

All important solution and simulation results are stored in structure `oo_`. A useful Octave command, showing the content of structure, is `fieldnames`. One can access any structure field by separating the name of the structure and its field by a dot. The most important fields of `oo_`:

- `oo_.dr`: model solution, including matrices of the recursive form for all model variables (with zero columns of the state transition matrix Φ suppressed)
- `oo_.gamma_y`: theoretical moments of model variables
- `oo_.mean`, `oo_.var`, `oo_.autocorr`: some basic moments from `oo_.gamma_y`
- `oo_.irfs`: impulse response functions

Further applications

The applications discussed above are merely a fraction of those implemented in Dynare, which also calculates e.g. higher order approximations, optimal policy analysis and estimation (including Bayesian methods).